

---

Subject: Re: Map\_grid options

Posted by [Kenneth P. Bowman](#) on Mon, 04 Oct 2004 00:15:40 GMT

[View Forum Message](#) <> [Reply to Message](#)

---

In article <dc38451.0410031406.3afdd995@posting.google.com>, jcesq@terra.com.br (Jeferson E.) wrote:

> I'm using /box\_axes option in the map\_grid procedure to draw axes  
> around my images. However, the text labels (lats and longs) are too  
> near to the axes. How can I move away these labels ?? The IDL's guide  
> doesn't show any option to do that.

My advice is to just accept that IDL does almost-publication-quality graphics. ;-)

Open the Postscript file in Illustrator and fix all the things you don't like.

Ken Bowman

---

---

Subject: Re: Map\_grid options

Posted by [jcesq](#) on Mon, 04 Oct 2004 17:50:46 GMT

[View Forum Message](#) <> [Reply to Message](#)

---

I have almost a thousand compositions to be done. I need some automactical procedure using map\_grid and box\_axes, or anything simililar :-)

Any suggsetions welcome

Jeferson

"Kenneth P. Bowman" <kpb@null.com> wrote in message news:<kpb-C357FF.19154003102004@news.tamu.edu>...

> In article <dc38451.0410031406.3afdd995@posting.google.com>,

> jcesq@terra.com.br (Jeferson E.) wrote:

>

>> I'm using /box\_axes option in the map\_grid procedure to draw axes  
>> around my images. However, the text labels (lats and longs) are too  
>> near to the axes. How can I move away these labels ?? The IDL's guide  
>> doesn't show any option to do that.

>

> My advice is to just accept that IDL does almost-publication-quality  
> graphics. ;-)

>

> Open the Postscript file in Illustrator and fix all the things you don't  
> like.

>  
> Ken Bowman

---

---

Subject: Re: Map\_grid options  
Posted by [K. Bowman](#) on Mon, 04 Oct 2004 19:27:55 GMT  
[View Forum Message](#) <> [Reply to Message](#)

---

In article <dc38451.0410040950.70d92a62@posting.google.com>, jcesq@terra.com.br (Jeferson E.) wrote:

> I have almost a thousand compositions to be done. I need some  
> automactical procedure using map\_grid and box\_axes, or anything  
> similar :-|

Ah, in that case, don't use BOX\_AXES. Also, don't use the automatic labels in MAP\_GRID. Draw the labels yourself with XYOUTS. (I also find that for cylindrical equidistant projections I need to draw the parallels and meridians myself (i.e., don't call MAP\_GRID).)

Unfortunately, you have to set things up specifically for each map projection that you use.

Something like this should get you started (dx and dy are fudge factors to get the label alignment right). This example is for a cylindrical equidistant global map.

Ken Bowman

PRO MAP\_TEST1

title = 'Map Test 1'

!P.POSITION = [0.1, 0.1, 0.9, 0.9]  
MAP\_SET, 0, 0, /CYLINDRICAL, LIMIT = [-90, -180, 90, 180], /NOBORDER  
MAP\_CONTINENTS

ch\_size = CONVERT\_COORD(!D.X\_CH\_SIZE, !D.Y\_CH\_SIZE, /DEVICE, /TO\_NORMAL)  
dx = ch\_size(0)  
dy = ch\_size(1)

xy = CONVERT\_COORD(0.0, 90.0, /DATA, /TO\_NORMAL)  
XYOUTS, xy(0), xy(1)+1.5\*dy, title, /NORMAL, ALIGNMENT = 0.5

xy0 = CONVERT\_COORD(-180.0, -90.0, /DATA, /TO\_NORMAL)  
xy1 = CONVERT\_COORD( 180.0, 90.0, /DATA, /TO\_NORMAL)

```

FOR xx = -180, 180, 90 DO BEGIN
  xy  = CONVERT_COORD(FLOAT(xx), -90.0, /DATA, /TO_NORMAL)
  PLOTS, [xy[0], xy[0]], [xy[1], xy[1]], PSYM = -3, /NORMAL
  xlabel = STRTRIM(STRING(xx), 2)
  XYOUTS, xy(0), xy(1)-2.0*dy, xlabel, /NORMAL, ALIGNMENT = 0.5
ENDFOR

FOR yy = -90, 90, 30 DO BEGIN
  xy  = CONVERT_COORD(-180.0, FLOAT(yy), /DATA, /TO_NORMAL)
  PLOTS, [xy[0], xy[0]], [xy[1], xy[1]], PSYM = -3, /NORMAL
  ylabel = STRTRIM(STRING(yy), 2)
  XYOUTS, xy(0)-dx, xy(1)-dy/2.0, ylabel, /NORMAL, ALIGNMENT = 1.0
ENDFOR

END

```

---

Subject: Re: Map\_grid options

Posted by [jcesq](#) on Tue, 05 Oct 2004 11:18:13 GMT

[View Forum Message](#) <> [Reply to Message](#)

---

OK! I'll try that!!

Thanks,  
J.

Kenneth Bowman <k-bowman@null.tamu.edu> wrote in message  
news:<k-bowman-7D66EA.14275504102004@news.tamu.edu>...

> In article <dc38451.0410040950.70d92a62@posting.google.com>,  
> jcesq@terra.com.br (Jeferson E.) wrote:

>

>> I have almost a thousand compositions to be done. I need some  
>> automactical procedure using map\_grid and box\_axes, or anything  
>> simlilar :-I

>

> Ah, in that case, don't use BOX\_AXES. Also, don't use the automatic  
> labels in MAP\_GRID. Draw the labels yourself with XYOUTS. (I also find  
> that for cylindrical equidistant projections I need to draw the  
> parallels and meridians myself (i.e., don't call MAP\_GRID).)

>

> Unfortunately, you have to set things up specifically for each map  
> projection that you use.

>

> Something like this should get you started (dx and dy are fudge factors  
> to get the label alignment right). This example is for a cylindrical  
> equidistant global map.

>

```

> Ken Bowman
>
>
> PRO MAP_TEST1
>
> title = 'Map Test 1'
>
> !P.POSITION = [0.1, 0.1, 0.9, 0.9]
> MAP_SET, 0, 0, /CYLINDRICAL, LIMIT = [-90, -180, 90, 180], /NOBORDER
> MAP_CONTINENTS
>
> ch_size = CONVERT_COORD(!D.X_CH_SIZE, !D.Y_CH_SIZE, /DEVICE, /TO_NORMAL)
> dx = ch_size(0)
> dy = ch_size(1)
>
> xy = CONVERT_COORD(0.0, 90.0, /DATA, /TO_NORMAL)
> XYOUTS, xy(0), xy(1)+1.5*dy, title, /NORMAL, ALIGNMENT = 0.5
>
> xy0 = CONVERT_COORD(-180.0, -90.0, /DATA, /TO_NORMAL)
> xy1 = CONVERT_COORD( 180.0,  90.0, /DATA, /TO_NORMAL)
>
> FOR xx = -180, 180, 90 DO BEGIN
>   xy   = CONVERT_COORD(FLOAT(xx), -90.0, /DATA, /TO_NORMAL)
>   PLOTS, [xy[0], xy[0]], [xy[1], xy[1]], PSYM = -3, /NORMAL
>   xlabel = STRTRIM(STRING(xx), 2)
>   XYOUTS, xy(0), xy(1)-2.0*dy, xlabel, /NORMAL, ALIGNMENT = 0.5
> ENDFOR
>
> FOR yy = -90, 90, 30 DO BEGIN
>   xy   = CONVERT_COORD(-180.0, FLOAT(yy), /DATA, /TO_NORMAL)
>   PLOTS, [xy[0], xy[0]], [xy[1], xy[1]], PSYM = -3, /NORMAL
>   ylabel = STRTRIM(STRING(yy), 2)
>   XYOUTS, xy(0)-dx, xy(1)-dy/2.0, ylabel, /NORMAL, ALIGNMENT = 1.0
> ENDFOR
>
> END

```

---